

Descripción BNF de la gramática de “TIMBA”

Las opciones escritas en rojo, preferentemente no deben utilizarse en la resolución de los ejercicios del práctico, a fin de acotar el conjunto de opciones y simplificar la escritura y la corrección de los programas.

<Programa timba> ::= <proceso> ; <declaraciones> .
 <proceso> ::= DEFINICION DE PROGRAMA <sentencias>
 <sentencias> ::= <sentencia> | <sentencias> , <sentencia>
 <sentencia> ::= <operativa> | <de control>
 <operativa> ::= <tomar> DE <pila> <nombre> | <depositar> EN <pila> <nombre> | <invertir>
 <tomar> ::= TOME UNA CARTA | TOME UNA | TOME
 <depositar> ::= DEPOSITE LA CARTA | DEPOSITELA
 <invertir> ::= INVIERTA LA CARTA | INVIERTALA
 <de control> ::= <de selección> | <iterativa>
 <de selección> ::= SI <condición> <sentencias> SINO <sentencias> NADA MAS | SI <condición> <sentencias> SINO NADA MAS
 <iterativas> ::= MIENTRAS <condición> <sentencias> REPITA
 <condición> ::= <condición simple> | <condición> Y <condición simple> | <condición> O <condición simple>
 <condición simple> ::= <condición de pila vacía> | <condiciones de carta>
 <condición de pila vacía> ::= <pila> <nombre> ESTA VACIA | <pila> <nombre> NO ESTA VACIA
 <condiciones de carta> ::= <estado> | <característica>
 <estado> ::= <carta> ESTA BOCA ABAJO | <carta> NO ESTA BOCA ABAJO
 <característica> ::= <carta> <es o no es> <de palos> | <carta> <es o no es> <valor> | <carta> ES DE <relac> PALO QUE TOPE DE <pila> <nombre>
 <es o no es> ::= ES | NO ES
 <carta> ::= LA CARTA | CARTA
 <pila> ::= LA PILA | PILA
 <de palos> ::= DEL PALO <palos> | <palos>
 <palos> ::= OROS | BASTOS | ESPADAS | COPAS
 <relac> ::= IGUAL | DISTINTO
 <valor> ::= DE <rela> VALOR QUE TOPE DE <pila> <nombre> | DE VALOR <relación> <número> | <relación> <número>
 <rela> ::= IGUAL | DISTINTO | MAYOR | MENOR | MAYOR O IGUAL | MENOR O IGUAL
 <relación> ::= IGUAL A | DISTINTO DE | MAYOR QUE | MENOR QUE | MAYOR O IGUAL A | MENOR O IGUAL A
 <número> ::= 1 | 2 | 3 | 4 | 5 | 6 | 7 | 10 | 11 | 12
 <declaraciones> ::= UCP EJECUTE CON LAS SIGUIENTES CARTAS : <lista de pilas>
 <lista de pilas> ::= <descripción de pila> | <lista de pilas> , <descripción de pila>
 <descripción de pila> ::= <pila> <nombre> <contenido>
 <contenido> ::= NO TIENE CARTAS | TIENE <lista de cartas>
 <lista de cartas> ::= <descripción de carta> | <lista de cartas> - <descripción de carta>
 <descripción de carta> ::= <número> DE <palos> | <número> DE <palos> ↑
 <nombre> es una cadena que no tiene blancos ni símbolos especiales de diez o menos caracteres alfabéticos o numéricos que no formen una palabra reservada.